

Textausgabe in Assembler

Das heutige Thema befasst sich mit einem zentralen Punkt des Operating-Systems:

Ich bin schon von vielen Leuten gefragt worden, wie man denn in Assembler "etwas auf den Bildschirm" bekommt - bitte schön - Sie können's im Anschluss lesen.

Leider ist die Ausgabe von Meldungen und Texten in Assembler nicht ganz so einfach wie in BASIC, welches über einen komfortablen "PRINT" Befehl verfügt. Kein Grund zur Verzweiflung, denn das Betriebssystem Ihres ATARI-Computers lässt Sie nicht im Stich.

Das Interface zum Betriebssystem wird durch das Unterprogramm TXTAUS hergestellt. Alles was Sie diesem Unterprogramm mitteilen müssen, ist die Anfangsadresse des auszudruckenden Textes, wobei der niederwertige Teil der Adresse (LSB) im Akku, der höherwertige Teil (MSB) im Y-Register übergeben wird.

Sehen Sie sich dazu das Demo-Programm (Zeilen 300-360) an, der Text selbst steht hinter einem .BYTE-Befehl in Anführungszeichen. Bei Eingabe von eigenen Texten sollten Sie niemals das EOL (RETURN) Zeichen \$9B am Ende des Textes vergessen. Tippen Sie das Listing mit Hilfe ihres Assemblers ein, bei MAC/65 und der Editor/Assembler Cartridge das .OPT OBJ in Zeile 90 beachten.

Jetzt assemblieren Sie den Quelltext und starten das Maschinenprogramm mit dem Debugger an der Adresse \$680 (Hexadezimal). Wenn Sie alles richtig gemacht haben, müsste die Meldung dann am Bildschirm stehen. Falls Ihr Assembler keinen eingebauten Debugger hat, können Sie das Programm auch im DOS starten, z.B. mit DOS-II Option M (Run at Address).

Noch ein paar Worte zur Funktion des TXTAUS Unterprogrammes. In den Zeilen 500 und 510 wird die übergebene Adresse in den IOCB Nummer 0 eingetragen. Sie wissen ja sicher, dass der IOCB Nr. 0 immer auf den Screen-Editor geöffnet ist. Die Zeilen 520 und 530 schreiben den Befehl "Put Text Record" in den IOCB, Zeile 540 bis 570 begrenzen die maximale Länge des Ausgabetextes auf 100 Zeichen. Die hier angegebene Länge sollte immer größer als die tatsächliche Länge des Textes sein, sonst wird er nicht vollständig ausgedruckt. Anschließend wird im X-Register eine Null hinterlegt, die dem nachfolgenden CIO-Aufruf mitteilt, dass der IOCB Nummer 0 zu bearbeiten ist. Sie bekommen von CIO im Y-Register eine Statusmeldung zurück, haben aber im vorliegenden Fall kaum Chancen, einen Fehler zu begehen.

Bleibt noch anzumerken, dass diese Methode der Textausgabe die universellste ist, da sie auf allen alten und neuen ATARIs funktioniert.

Peter Finzel (1985)

```

0000          90      .OPT OBJ
0100 ;*****
0110 ;Textausgabe in Assembler fuer alle ATARI-Computer
0120 ;
0130 ;Peter Finzel 1984
0140 ;*****
0150 ;
0160 ;Einige Adressen des Betriebesystemes:
0170 ;
=0009      0180 CPTXTR = $09      ; CIO-Befehlscode fuer Put Text Record
=0342      0190 ICCOM = $0342    ; IOCB Nr.0 Kommandoregister
=0344      0200 ICBADR = $0344   ; Bufferadresse IOCB#0
=0348      0210 ICBLN = $0348   ; Laenge dieses Buffers
=E456      0220 CIOV = $E456    ; CIO Einsprungsadresse
0230 ;
0240 ;*****
0250 ;Zuerst ein Demoprogramm das Ihnen zeigt,
0260 ;wie TXTAUS benutzt wird:
0270 ;*****
0280 ;
0000      0290      *= $0680    ; in PAGE 6, zweite Haelfte
0300 ;
0680 A988      0310 DEMO LDA #TEXT1&255 ; LSB des Zeigers in Akku
0682 A006      0320      LDY #TEXT1/256 ; MSB in Y-Register
0684 20A406     0330      JSR TXTAUS  ; und ausgeben.
0687 60         0340      RTS        ; das war's!
0350 ;
0688 44696573   0360 TEXT1 .BYTE "Diese Zeile wird ausgegeben", $9B
068C 65205A65
0690 696C6520
0694 77697264
0698 20617573
069C 67656765
06A0 62656E9B

0380 ;*****
0390 ;Das Textausgabe Unterprogramm:
0400 ;
0410 ;So benutzen Sie TXTAUS:
0420 ; -Akku/Y-Register mit der Adresse des Textes laden
0440 ; (Akku mit LSB, Y mit MSB)
0450 ; -JSR TXTAUS ausfuehren
0480 ;*****
0490 ;
06A4 8D4403     0500 TXTAUS STA ICBADR ; Adresse LSB in IOCB#0 eintragen
06A7 8C4503     0510      STY ICBADR+1 ; jetzt ist MSB dran
06AA A909       0520      LDA #CPTXTR ; Befehl: Textausgabe
06AC 8D4203     0530      STA ICCOM  ; ebenfalls in IOCB
06AF A964       0540      LDA #100  ; max. Laenge des Textes vorgeben
06B1 8D4803     0550      STA ICBLN ; hier = 100 (willkuerlich gewaehlt)
06B4 A900       0560      LDA #0    ; MSB Laenge ist Null
06B6 8D4903     0570      STA ICBLN+1
06B9 A200       0580      LDX #0    ; IOCB Nr.0 ist gemeint
06BB 2056E4     0590      JSR CIOV  ; Zentrale I/O Routine aufrufen
06BE 60         0600 ;          Eventl. Fehlercode ist im Y-Register
0620      RTS

ASSEMBLY ERRORS: 0   31626 BYTES FREE

```

Bild1: Demo-Listing im Atari-Assembler Format

```
*****
* TEXTAUSGABE IN ASSEMBLER (ATMA5-II)
* PETER FINZEL 1984
*****
```

```
CPTXTR EQU $09          CIO Befehl „put text“
ICCOM EQU $0342         Kommandobyte IOCB 0
ICBADR EQU $0344        Puffer-Adresse
ICBLEN EQU $0348        Puffer-Länge
CIOV EQU $E456          CIO-Einsprungsadresse
```

```
ORG $0680
```

```
*****
* DEMOPROGRAMM ZUR NUTZUNG VON TEXTAUS
*****
```

```
DEMO LDA #TEXT1:L      LSB von Text1 in Akku
      LDY #TEXT1:H      MSB in Y-Register
      JSR TXTAUS        Ausgaberroutine aufrufen
      RTS              ENDE
TEXT1 ASC "Diese Zeile wird ausgegeben"
      DFB $9B
```

```
*****
* TEXTAUSGABE UNTERBROGRAMM
*
* AN AKKU und Y TEXTADRESSE UEBERGEHEN
* A=LSB Y=MSB
* TEXT MUSS MIT $9B (EOF) ENDEN
*****
```

```
TXTAUS STA ICBADR       TXTADRESSE an IOCB
        STY ICBADR+1
        LDA #CPTXTR     BEFEHL für PUT TEXT
        STA ICCOM       an IOCB
        LDA #100        MAXIMALE TEXTLAENGE
        STA ICBLEN
        LDA #0
        STA ICBLEN+1
        LDX #0          IOCB#0
        JSR CIOV        IOCB ROUTINE AUFRUFEN
        RTS
```